X-671-64-334

NASA TM X-55149

N65-18190

22 (PAGES)

TMX-55149 (NASA CR OR TMX OR AD NUMBER)

(ACCESSION NUMBER) (THRU)

1 (CODE)

08 (CATEGORY)

FACILITY FORM 602

AVOIDING "BLACK BOX" COMPUTER PROGRAMS

BY

ABROM HISLER

GPO PRICE $ ______

OTS PRICE(S) $ ______

Hard copy (HC) $1.00

Microfiche (MF) $0.50

OCTOBER 1964

GODDARD SPACE FLIGHT CENTER

GREENBELT, MARYLAND

AVOIDING "BLACK BOX" COMPUTER PROGRAMS

by

Abrom Hisler

Goddard Space Flight Center

SUMMARY

18190

Too many useful computer programs go the way of undocumented "black boxes" which are treacherous to run. A computer report format is set forth to salvage such programs with a minimum of effort. It is based on the "Liquan" program designed to analyze liquid-propellant rocket-engine static firing data. A feature to readily accommodate program expansion is also presented–the engineering analysis sequence pattern.

Author

CONTENTS

AVOIDING "BLACK BOX" COMPUTER PROGRAMS

by

Abrom Hisler

Goddard Space Flight Center

INTRODUCTION

No sooner does an engineering computer program become operational than the engineer-programmer team, responsible for the program's development, must seriously consider the need to document the effort which went into the program. To do otherwise is to risk the chameleon conversion of the new useful program into a treacherous "black box."

Yet, too often computer programs, once developed and checked out, are never documented or are not sufficiently well documented. Such programs run the risk of becoming "black boxes" if:

1. The original engineer-programmer team is not available for consultation.
2. Through the years, this original team has lost their intimate familiarity with the program's details.

Periodic critiques of "black box" programs can be undertaken only with great difficulty. Without the documented development effort available, updating the program becomes impossible.

The engineer-programmer team's interest, initially keen to develop a particular computer program to meet certain needs, often wanes once the program has been developed and attention shifts elsewhere to develop new programs.

"Black box" programs are treacherous to run because the original assumptions, unique conditions of the computation, and other program idiosyncrasies have become unknown "booby trap" quantities. The cost of operating such programs grows to include not only the actual computer time wasted but also the unwise expenditure of money on engineering designs which reflect an erroneous conception of the computer program employed.

Constructive critiques of "black box" programs are impossible and so is the possibility of updating these programs as new state-of-the-art techniques are developed. A well-documented computer program reflects the developmental effort which went into the program, permits effective periodic critiques, and can readily accommodate new calculation methods.

The nature of the computer program documentation considered in this report differs from other similar reports (Ref. 1) in presenting a few features which should appeal more to the engineer collaborating with his technical writer-editor than to the programmer. The programmer, should he feel so inclined, might wish to prepare a computer flow chart (Ref. 1, p. 37), which is usually more comprehensible to programmers rather than engineers.

PREFERABLE FORMATS

The need to establish a preferable format for a well-documented report on computer programs was first pointed out at the Ninth Meeting of the Working Group on Design Automation (Ref. 2, p. 24). Although there is nothing startling about the "Liquan" computer program documented in this report, it will serve mainly to present the essentials of this "preferable format." They include the following:

1. Computer program
2. Sequential engineering analyses
3. Decoding and worksheet
4. Typical input
5. An additional input format
6. Typical output

SEQUENTIAL ENGINEERING ANALYSES

Although a copy of the computer program is presented in Appendix A, reference to the Sequential Engineering Analysis of Appendix B should more quickly familiarize the reader with the essence of the program. The equations presented are of general utility; but many of the constants, such as the 4.6 in the coefficient of thrust equation, refer specifically to the Aerobee 350 sounding rocket—in this case, the ratio of exit area to throat area of the nozzle. A subsequent follow-on program sought to empirically relate the C* function in terms of mixture ratio and chamber pressure for later use in an associated propellant utilization program, for still subsequent use in a simple 2-D trajectory program.

A number of additional features present in Appendix A but not shown in Appendix B include:

1. TC – The time at which a particular chamber changes from step operation to full operation. This permits a change in the K_W values so that

the propellant flow is markedly increased. The step operation stage is employed to avoid the problem of feeding too much propellant into the chamber and igniting it after a significant ignition delay period. Were this to happen, the chamber would blow up.

2. TFIN and TFOUT — The fuel temperature, in and out of the cooling jacket, is averaged before its specific gravity is computed.

3. A CFA feature to throw out negative CFA values where the chamber pressure has not risen sufficiently to even reach the step operation stage.

These features have not been incorporated into the Sequential Engineering Analysis to keep it sufficiently simple for presentation in this report. However, these features (and others) would take their respective place in the analysis in a followup revision as part of the updating effort.

It can therefore be seen that, although Appendix B does not reflect every detail of the present program, it can nevertheless readily accommodate additions and deletions to the program and so grow in complexity but not in confusion.

Appendix B can also be used as a worksheet to jot down references in their appropriate place in the sequence for future inclusion of this matter into the program.

It also permits simpler constructive critique as nebulous areas arise or as the conditions, circumstances, and assumptions of the program computation change.

Appendix B also serves as a pivot point. Its analysis can be pivoted in the direction of any of the program languages—FORTRAN, MYSTIC, FORAST, etc. And, conversely, these computer languages can be translated back into the "engineering" language of Appendix B.

DECODING WORKSHEET

The Fortran Decoding and Worksheet in Appendix C neatly ties together in one chart:

1. The input and output,
2. The engineering notation with dimensions employed in the Sequential Engineering Analysis,

3. The blank three columns of card numbers, column numbers, and typical values which may be used by the program operator in the manner of a worktable for better familiarization with the input of the program.

It may be appropriate to mention here with respect to engineering notation and FORTRAN that a step has been taken to standardize FORTRAN nomenclature as it pertains to solid rocket motor design (Ref. 3).

DATA INPUT

The data input is presented in two forms: Appendixes D and E. The usual order of presentation is shown in the former and, in the latter, a new method associates the read input tape card with corresponding data. The latter appendix should reduce the time required to prepare new sets of data, especially since the data are aligned with the numbered columns at the bottom of the sheet. Another advantage of the Appendix E input is to eliminate costly errors which occur when misunderstandings arise over the data input. Too often, useless program runs are attempted, resulting in wasteful computer time expenditures and lost man-hours because of initial unfamiliarity with data input order.

Typical output is shown in Appendix F.

FORMAT ADVANTAGES

With the above formats, less effort is required to document a particular computer program; and the computer report so prepared has the following advantages:

1. Presents a bird's-eye view of the calculation logic and flow.
2. Affords the opportunity to annotate the engineering analysis flow chart with:
   (a) Assumptions,
   (b) Unique conditions,
   (c) Past and future applicable references.
3. Permits a fast periodic review of the program.

4. Involves a minimum of time and effort to familiarize, and refamiliarize oneself after a long lapse of time, with the program flow and data input.
5. Serves to expand the original program as the occasion arises.

RECAPITULATION

This computer report then serves two purposes:

1. To provide the necessary information to utilize the "Liquan" computer program to ANalyze LIQUid (LIQU-AN) propellant rocket-engine static firing data.
2. To set forth a computer report format of general applicability which may differ markedly from typical computer program reports but which contains the minimum necessary essentials to adequately document a computer program, such as "Liquan."

REFERENCES

1. Frankle, J. M., "The Effect of Distribution of Impulse on Range for a Gun-Boosted Rocket," BRL Memorandum Report No. 1548, February 1964, Aberdeen Proving Ground, Maryland.

2. Proceedings of the Ninth Meeting, Interagency Chemical Rocket Propulsion Working Group on Design Automation, CPIA Publication No. 55, August 1964, Chemical Propulsion Information Agency, The Johns Hopkins University Applied Physics Laboratory, Silver Spring, Maryland.

3. Interagency Chemical Rocket Propulsion Working Group on Design Automation, "Computer Program Nomenclature Standardization," CPIA Publication No. 28, October 1963, Chemical Propulsion Information Agency, The Johns Hopkins University Applied Physics Laboratory, Silver Spring, Maryland.

APPENDIX A

THE COMPUTER PROGRAM—LIQUAN

Appendix A

The Computer Program—Liquan

```
$JOB   1091C003 409HISLER, A.
$EXECUTE        FORTRA
*
*      XEQ
*      XEQ
*      LIST 8
*      LABEL
C LIQUAN-A PRELIMINARY ANALYSIS OF LIQUID PROPELLANT ROCKET ENGINE
C           STATIC FIRING DATA-A.HISLER                                    4
      DIMENSION PC(200),CFA(200),TIME(200),PCDT(200),PCG(200),SPGFA(200)
     1,TFA(200),DPFA(200),PFJ(200),WDFA(200),DPO(200),POJ(200),WDO(200),
     2WDT(200),WDFDT(200),WDODT(200),WDTDT(200),DELT(200),AMR(200),PALT(
     3200),FA(200),AISP(200),CSTAR(200),TFIN(200),TFOUT(200)
   10 AIT=0.0
      TWDTDT=0.0
      TWDODT=0.0
      TWDFDT=0.0
      READ INPUT TAPE 2,1,IA,IB,IC,ID,IE
    1 FORMAT(I4,4I2)
      READ INPUT TAPE 2,2,AT,TO,AKWFA,AKWO,TKWFA,TKWO,TC
    2 FORMAT(7F10.4)
      READ INPUT TAPE 2,14,G
   14 FORMAT(F10.4)
      READ INPUT TAPE 2,3,(TIME(I), I=1,IA)
      READ INPUT TAPE 2,3,(TFIN(I),I=1,IA)
      READ INPUT TAPE 2,3,(TFOUT(I),I=1,IA)
      READ INPUT TAPE 2,3,(PFJ(I), I=1,IA)
      READ INPUT TAPE 2,3,(POJ(I), I=1,IA)
      READ INPUT TAPE 2,3,(PCG(I),I=1,IA)
    3 FORMAT(7F10.4)
      SPGO=-0.000833*TO+1.5953
      WRITE OUTPUT TAPE 3,36
   36 FORMAT(1H1)
      WRITE OUTPUT TAPE 3,47
   47 FORMAT(49H WATER CONSTANTS   THROAT AREA    G-VALUE   OXIDIZER/51H
     1FUEL    OXIDIZER                          TEMPERATURE)
      WRITE OUTPUT TAPE 3,4,AKWFA,AKWO,AT,G,TO
    4 FORMAT(5F10.4)
      WRITE OUTPUT TAPE 3,15
   15 FORMAT(1H0)
      WRITE OUTPUT TAPE 3,8
    8 FORMAT(1H0/105H                    FLOW RATE                    MIXTURE
     1  CF          THRUST       SPECIFIC      CSTAR            CHAMBER/116H     OXID
     2IZER         FUEL          TOTAL        RATIO
     3 IMPULSE                         PRESSURE          TIME)
      DO 11 I=1,IA
      TFA(I)=0.5*(TFIN(I)+TFOUT(I))
      SPGFA(I)=-0.000508*TFA(I)+1.0988
      DPFA(I)=PFJ(I)-PCG(I)
      DPO(I)=POJ(I)-PCG(I)
      IF (PCG(I)) 81,81,82
   81 PC(I)=0.0
      GO TO 83
   82 PC(I)=PCG(I)+14.7
   83 GO TO (7,5),IB
    7 READ INPUT TAPE 2,2,(PALT(J), J=1,IA)
    6 CFA(I)=1.422+(0.036-(PALT(I)/PC(I)))*4.6
      GO TO 12
    5 CFA(I)=1.422+(0.036-14.7/PC(I))*4.6
      IF(CFA(I)) 21,12,12
   21 CFA(I)=0.0
   12 FA(I)=CFA(I)*AT*PC(I)
      IF(TIME(I)-TC) 70,60,60
   70 WDFA(I)=TKWFA*SQRTF(DPFA(I)*SPGFA(I))
      WDO(I)=TKWO*SQRTF(DPO(I)*SPGO)
      GO TO 80
   60 WDFA(I)=AKWFA*SQRTF(DPFA(I)*SPGFA(I))
      WDO(I)= AKWO*SQRTF(DPO(I)*SPGO)
   80 WDT(I)=WDFA(I)+WDO(I)
      AMR(I)=WDO(I)/WDFA(I)
      AISP(I)=FA(I)/WDT(I)
      CSTAR(I) =(PC(I)*AT*G)/WDT(I)
   11 WRITE OUTPUT TAPE 3,9,WDO(I),WDFA(I),WDT(I),AMR(I),CFA(I),FA(I),AI
     1SP(I),CSTAR(I),PC(I),TIME(I)
    9 FORMAT(F10.4,2F12.4,F10.4,F12.4,F14.4,3F12.4,F11.4)
      WRITE OUTPUT TAPE 3,15
      IA1=IA-1
      DO 50 I=1,IA1
      DELT(I)=TIME(I+1)-TIME(I)
      CALL TRAPI(PC(I+1),PC(I),DELT(I),PCDT(I))
      AIT=AIT+CFA(I)*AT*PCDT(I)
      CALL TRAPI(WDO(I+1),WDO(I),DELT(I),WDODT(I))
      TWDODT=TWDODT+WDODT(I)
      CALL TRAPI(WDFA(I+1),WDFA(I),DELT(I),WDFDT(I))
      TWDFDT=TWDFDT+WDFDT(I)
      CALL TRAPI(WDT(I+1),WDT(I),DELT(I),WDTDT(I))
   50 TWDTDT=TWDTDT+WDTDT(I)
      WRITE OUTPUT TAPE 3,41,IC
   41 FORMAT(30H 350 AEROBEE STATIC FIRING NO I2)
      WRITE OUTPUT TAPE 3,42,ID
   42 FORMAT(24H 350 AEROBEE CHAMBER NO I2)
      WRITE OUTPUT TAPE 3,43,IE
   43 FORMAT(33H AEROBEE PRESSURE DROP SYSTEM NO I2)
      WRITE OUTPUT TAPE 3,44
   44 FORMAT(39H  AEROBEE 350 PRESSURE DROP SYSTEM CODE/33H       NO 1-INJ
     1ECTION PRESSURE DROP/38H       NO 2-TANK PRESSURE DROP
      WRITE OUTPUT TAPE 3,15
      WRITE OUTPUT TAPE 3,13,AIT
   13 FORMAT(37H THE TOTAL IMPULSE OF THIS CHAMBER = F10.2)
      WRITE OUTPUT TAPE 3,16,TWDFDT
   16 FORMAT(36H THE TOTAL WEIGHT OF FUEL FLOWED IS F10.4)
      WRITE OUTPUT TAPE 3,17,TWDODT
   17 FORMAT(40H THE TOTAL WEIGHT OF OXIDIZER FLOWED IS F10.4)
      WRITE OUTPUT TAPE 3,18,TWDTDT
   18 FORMAT(42H THE TOTAL WEIGHT OF PROPELLANT FLOWED IS F10.4)
      TOTP=TWDFDT + TWDODT
      WRITE OUTPUT TAPE 3,19,TOTP
   19 FORMAT(53H THE COMBINED FUEL AND OXIDIZER PROPELLANT FLOWED IS F10
     1.4)
      GO TO 10
      END
*      LIST 8
*      LABEL
C  TRAPEZOIDAL INTEGRATION
      SUBROUTINE TRAPI(B,C,D,A)
      A=(B+C)*0.5*D
      RETURN
      END
```

Appendix B

Sequential Engineering Analysis

$$C_f = 1.422 + \left[0.036 - \left(P_{alt}/P_c\right)\right] 4.6$$

$$F = C_f\, P_c\, \Delta t$$

$$I_T = C_f\, A_t \int P_c\, \Delta t$$

$$\int P_c\, \Delta_t = \sum \frac{1}{2}\left[\left(P_c\right)_i + \left(P_c\right)_{i+1}\right] \Delta t$$

$$\dot{w}_f = K_w \sqrt{\Delta P_f\, \rho_f}$$

$$\rho_f = -\,0.000508\, T_f + 1.0988$$

$$\Delta P_f = P_f - P_c$$

$$I_{SP} = F/\dot{w}_t$$

$$C^* = \frac{P_c\, A_t\, g}{\dot{w}_T}$$

$$\dot{w}_T = \dot{w}_f + \dot{w}_o$$

$$MR = \frac{\dot{w}_o}{\dot{w}_f}$$

$$\dot{w}_o = K_w \sqrt{\Delta P_o\, \rho_o}$$

$$\rho_o = -\,0.000833\, T_o + 1.5953$$

$$\Delta P_o = P_o - P_c$$

$$w_T = \sum \frac{1}{2}\left[\left(\dot{w}_T\right)_i + \left(\dot{w}_T\right)_{i+1}\right] \Delta t$$

$$w_o = \sum \frac{1}{2}\left[\left(\dot{w}_o\right)_i + \left(\dot{w}_o\right)_{i+1}\right] \Delta t$$

$$w_f = \sum \frac{1}{2}\left[\left(w_f\right)_i + \left(w_f\right)_{i+1}\right] \Delta t$$

Reference: Zucrow, M. J., Aircraft and Missile Propulsion, Vol. II, Chapt. 10, John Wiley and Sons, Inc., 1958.

Appendix C

Fortran Decoding and Worksheet

Fortran Code			Notation	Definition	Card No.	Column Nos.	Dimension	Typical Value
In	Neither	Out						
		AMR		Mixture ratio				
		AISP	I_{SP}	Specific impulse			lb/(lb/sec)	
		AIT	I_T	Total impulse			lb sec	
AT			A_t	Area of thrust cross section of discharge nozzle			$in.^2$	
AKWFA			$(k_w)_f$	Water flow constant for fuel system–steady state				
AKWO			$(k_w)_o$	Water flow constant for oxidizer system–steady state				
		CFA	C_F	Thrust coefficient = $F/A_t\ P_c$				
		CSTAR	C^*	Characteristic velocity for rocket motor			ft/sec	
	DPFA		ΔP_f	Pressure differential across fuel system			psi	
	DPO		ΔP_o	Pressure differential across oxidizer system			psi	
	DELT		Δt	Time interval			sec	
		FA	F	Thrust			lb	
G			g	Acceleration due to gravity				
IA				No. of data points				
IB				Do either statement 7(1) or 5(2)				
IC				Firing no.				
ID				Chamber no.				
IE				Pressure drop system no.				
	IA1			No. of data points - 1				
		PC	P_c	Chamber pressure, absolute			psia	
	PCDT		$P_c \Delta t$	Chamber pressure time integral				
PCG			P_{cg}	Chamber pressure, gage			psig	
PFJ			P_{fj}	Fuel injection pressure				
POJ			P_{oj}	Oxidizer injection pressure				
PACT			P_{alt}	Pressure at altitude				
	SPGFA		ρ_f	Specific gravity of fuel				
	SPGO		ρ_o	Specific gravity of oxidizer				
TIME			t	Time				
	TFA		T_f	Temperature of fuel, average			°F	
TFIN			T_{fi}	Incoming temperature of fuel			°F	
TFOUT			T_{fo}	Exit temperature of fuel			°F	
		TWDTDT		Total flow rate-time integral				
		TWDODT		Total oxidizer flow rate-time integral				
		TWDFDT		Total fuel flow rate-time integral				
TO			T_o	Oxidizer temperature, average			°F	
TKWFA				Initial transient fuel water flow constant				
TKWO				Initial transient oxidizer water flow constant				
TC			t_c	Time of propellant full flow				
		WDFA	$\dot{w}_f$	Fuel flow rate			lb/sec	
		WDO	$\dot{w}_o$	Oxidizer flow rate			lb/sec	
		WDT	$\dot{w}_T$	Total propellant flow rate			lb/sec	
	WDFDT			Fuel flow rate–time integrals				
	WDODT			Oxidizer flow rate–time integrals				
	WDTDT			Propellant's flow rate–time integrals				

Appendix D

Typical Data Input

```
*      DATA
 17  2  3  1  1
   9.208     91.0       0.503      1.38       0.146      0.412      1.581
  32.2
   1.414      1.504      1.645      1.811      2.029      3.065      4.064
  53.011     53.203     53.408     53.600     53.805     54.010     54.509
  55.008     55.405     55.815
  79.479     79.034     79.523     80.057     77.877    108.032    163.912
 230.587    239.253    244.973    247.919    236.220    227.554    500.256
 196.474    190.460    181.868
  79.479     79.034     79.523     80.057     77.877    108.032    163.912
 236.567    231.887    231.887    233.880    232.754    231.454    230.587
 196.474    190.460    181.868
 412.672    427.755    375.817    386.541    387.819    385.166    387.268
 356.015    335.425    309.990    301.264    326.630    290.282    232.298
 190.961    165.750    146.717
  33.358    106.196    353.787    359.720    367.686    363.893    362.144
 386.636    381.887    384.386    385.086    370.791    381.287    382.587
 374.290    370.591    375.239    372.170    359.725    357.876    356.121
    .119     68.535    290.466    297.122    301.264    296.048    296.398
 288.233    251.607    203.832    105.293     14.638     11.751      2.743
   1.350       .469       .443
```

Appendix E

An Additional Data Input Format

```
*     DATA
      READ INPUT TAPE 2,1,IA,IB,IC,ID,IE
 17 2 3 1 1
      READ INPUT TAPE 2,2,AT,TO,AKWFA,AKWO,TKWFA,TKWO,TC
  9.208     91.0      0.503     1.38      0.146     0.412     1.581
      READ INPUT TAPE 2,14,G
  32.2
      READ INPUT TAPE 2,3,(TIME(I), I=1,IA)
  53.011    53.203    53.408    53.600    53.805    54.010    54.509
      READ INPUT TAPE 2,3,(TFIN(I),I=1,IA)
 230.587   239.253   244.973   247.919   236.220   227.554   500.256
      READ INPUT TAPE 2,3,(TFOUT(I),I=1,IA)
 236.567   231.887   231.887   233.880   232.754   231.454   230.587
      READ INPUT TAPE 2,3,(PFJ(I), I=1,IA)
 356.015   335.425   309.990   301.264   326.630   290.282   232.298
      READ INPUT TAPE 2,3,(POJ(I), I=1,IA)
 386.636   381.887   384.386   385.086   370.791   381.287   382.587
      READ INPUT TAPE 2,3,(PCG(I),I=1,IA)
 288.233   251.607   203.832   105.293    14.638   -11.751     2.743

1234567890
        1-1234567890
                  2-1234567890
                            3-1234567890
                                      4-1234567890
                                                5-1234567890
                                                          6-1234567890
```

Appendix F

Typical Data Output

WATER CONSTANTS FUEL	WATER CONSTANTS OXIDIZER	THROAT AREA	G-VALUE	OXIDIZER TEMPERATURE
0.5030	1.3800	9.2080	32.2000	91.0000

FLOW RATE OXIDIZER	FLOW RATE FUEL	TOTAL	MIXTURE RATIO	CF	THRUST	SPECIFIC IMPULSE	CSTAR	CHAMBER PRESSURE	TIME
2.9280	3.0509	5.9789	0.9597	0.	0.	0.	734.8885	14.8190	1.4140
3.1167	2.8471	5.9638	1.0947	0.7752	594.1359	99.6232	4138.1086	83.2350	1.5040
13.5364	4.7808	18.3172	2.8314	1.3660	3838.4609	209.5555	4939.6844	305.1660	1.6450
13.4589	4.8927	18.3516	2.7508	1.3707	3935.7625	214.4639	5037.9420	311.8220	1.8110
13.8639	4.8163	18.6802	2.8786	1.3736	3996.3128	213.9336	5015.0845	315.9640	2.0290
14.0116	4.8516	18.8632	2.8880	1.3700	3920.0620	207.8153	4884.4332	310.7480	3.0650
13.7932	4.8320	18.6251	2.8546	1.3702	3925.1786	210.7463	4952.4363	311.0980	4.0640
16.8746	4.0999	20.9745	4.1159	1.3644	3805.8175	181.4501	4282.2996	302.9330	53.0110
19.4164	4.5568	23.9731	4.2610	1.3337	3270.3959	136.4192	3293.6617	266.3070	53.2030
22.8577	5.1244	27.9821	4.4606	1.2782	2571.9914	91.9156	2315.5600	218.5320	53.4080
28.4543	6.9580	35.4122	4.0894	1.0241	1131.4872	31.9519	1004.6705	119.9930	53.6000
32.1031	8.7939	40.8970	3.6506	0.	0.	0.	212.6963	29.3380	53.8050
32.7007	8.3197	41.0204	3.9305	0.	0.	0.	191.1891	26.4510	54.0100
33.1537	7.2826	40.4363	4.5525	0.	0.	0.	127.9002	17.4430	54.5090
32.8510	6.9228	39.7738	4.7453	0.	0.	0.	119.6463	16.0500	55.0080
32.7266	6.4733	39.1999	5.0557	0.	0.	0.	114.7343	15.1690	55.4050
32.9326	6.1029	39.0356	5.3962	0.	0.	0.	115.0198	15.1430	55.8150

350 AEROBEE STATIC FIRING NO 3
350 AEROBEE CHAMBER NO 1
AEROBEE PRESSURE DROP SYSTEM NO 1
AEROBEE 350 PRESSURE DROP SYSTEM CODE
NO 1-INJECTION PRESSURE DROP
NO 2-TANK PRESSURE DROP

THE TOTAL IMPULSE OF THIS CHAMBER = 201155.76
THE TOTAL WEIGHT OF FUEL FLOWED IS 250.1291
THE TOTAL WEIGHT OF OXIDIZER FLOWED IS 870.5084
THE TOTAL WEIGHT OF PROPELLANT FLOWED IS 1120.6375
THE COMBINED FUEL AND OXIDIZER PROPELLANT FLOWED IS 1120.6375